

Salesforce Marketing Cloud

AMPscript Loops

Exercise 2



Overview

This exercise is designed to test your understanding of For Loops in AMPscript.

Process the tasks in order to complete this exercise.

Each Task in the following Brief will get progressively harder, so if you get stuck or are unsure how to proceed, you can join the conversation in the comments section here: <https://youtu.be/rNJlVNajTmU>

The last pages of this document contain some solutions. Make sure you try solve them yourself before jumping to the end.

For Loops are an incredibly powerful tool and they can be used to make highly dynamic emails for your customers. I hope this exercise has helped to strengthen your knowledge and given you some ideas on how to use For Loops in your email sends!

Required Files

This exercise can be completed on a Cloud Page or in an Email. It does not require any external data.

You can find an up to date copy of this exercise brief here:

<https://github.com/camrobert/SalesforceMarketingCloud/tree/main/Challenge/AMP-E02>



Email Setup

- 1 In Content Builder, Create an email using the "Blank Page" template. Name your email "AMP-E02 <yourname>".

Alternative Option: You can also choose to complete this task on a Cloud Page, as there is no subscriber data being used.

Each task builds on the solution from the previous, so you can choose to attempt each task by altering the previous solution, or by creating a new content block each time.

Creating a new content block each time will allow you to keep copies of your code through each task in this exercise.



Task 1

Using an AMPscript For Loop, create a loop which outputs the numbers from 1 to 10, with each number occurring on a new line.

Expected Results

1
2
3
4
5
6
7
8
9
10



Task 2

Alter your code so the For Loop outputs the word "Even" when the number is even, and outputs the number when it is odd.

Expected Results

```
1
Even
3
Even
5
Even
7
Even
9
Even
```



Task 3

Alter your code so that it loops for 100 cycles and implement the rules for the game FizzBuzz:

- Numbers divisible by 3: Output "Fizz"
- Numbers divisible by 5: Output "Buzz"
- Numbers divisible by 3 & 5: Output "FizzBuzz"
- Otherwise, output the number

Each result should occur on a new line.

Reference: https://en.wikipedia.org/wiki/Fizz_buzz

Video Reference: https://www.youtube.com/watch?v=QPZ0pIK_wsc

Expected Results

```
1
2
Fizz
4
Buzz
Fizz
7
8
Fizz
Buzz
11
...
```



Task 4

Keeping your output values as they are, change your code so that the values are printed into a 10x10 table.

Give your table the "border=1" property so that you can see the cells.

Expected Result

| | | | | | | | | | |
|------|------|------|------|----------|------|------|------|------|----------|
| 1 | 2 | Fizz | 4 | Buzz | Fizz | 7 | 8 | Fizz | Buzz |
| 11 | Fizz | 13 | 14 | FizzBuzz | 16 | 17 | Fizz | 19 | Buzz |
| Fizz | 22 | 23 | Fizz | Buzz | 26 | Fizz | 28 | 29 | FizzBuzz |
| 31 | 32 | Fizz | 34 | Buzz | Fizz | 37 | 38 | Fizz | Buzz |
| 41 | Fizz | 43 | 44 | FizzBuzz | 46 | 47 | Fizz | 49 | Buzz |
| Fizz | 52 | 53 | Fizz | Buzz | 56 | Fizz | 58 | 59 | FizzBuzz |
| 61 | 62 | Fizz | 64 | Buzz | Fizz | 67 | 68 | Fizz | Buzz |
| 71 | Fizz | 73 | 74 | FizzBuzz | 76 | 77 | Fizz | 79 | Buzz |
| Fizz | 82 | 83 | Fizz | Buzz | 86 | Fizz | 88 | 89 | FizzBuzz |
| 91 | 92 | Fizz | 94 | Buzz | Fizz | 97 | 98 | Fizz | Buzz |



Task 5

Keeping the FizzBuzz game logic, change your 10x10 table so that it outputs the numbers from 1-100, and the background of each cell reflects the rules of FizzBuzz as follows:

- Fizz = "Red"
- Buzz = "Green"
- FizzBuzz = "Blue"
- (none) = "White"

Keep the "border=1" property, and force each cell to be 20px in height and width so the table forms a square.

Expected Result

| | | | | | | | | | |
|----|----|----|----|----|----|----|----|----|-----|
| 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9 | 10 |
| 11 | 12 | 13 | 14 | 15 | 16 | 17 | 18 | 19 | 20 |
| 21 | 22 | 23 | 24 | 25 | 26 | 27 | 28 | 29 | 30 |
| 31 | 32 | 33 | 34 | 35 | 36 | 37 | 38 | 39 | 40 |
| 41 | 42 | 43 | 44 | 45 | 46 | 47 | 48 | 49 | 50 |
| 51 | 52 | 53 | 54 | 55 | 56 | 57 | 58 | 59 | 60 |
| 61 | 62 | 63 | 64 | 65 | 66 | 67 | 68 | 69 | 70 |
| 71 | 72 | 73 | 74 | 75 | 76 | 77 | 78 | 79 | 80 |
| 81 | 82 | 83 | 84 | 85 | 86 | 87 | 88 | 89 | 90 |
| 91 | 92 | 93 | 94 | 95 | 96 | 97 | 98 | 99 | 100 |



Answers



Answers

Note: Many of these Tasks can be solved multiple ways using AMPscript. The Answers below are not exclusively correct, however they will work if copied into your email for this exercise.

Answers are shown in HTML for formatting purposes.

1

```
%%[FOR @i = 1 TO 10 DO  
Output(Concat(@i,"<br>"))  
NEXT @i]%%
```

2

```
%%[FOR @i = 1 TO 10 DO  
IF MOD(@i,2) == 0 THEN]%%  
Even <br>  
%%[ELSE  
Output(Concat(@i,"<br>"))  
ENDIF  
NEXT @i]%%
```

3

```
%%[FOR @i = 1 TO 100 DO  
IF MOD(@i,15) == 0 THEN SET @output = "FizzBuzz"  
ELSEIF MOD(@i,5) == 0 THEN SET @output = "Buzz"  
ELSEIF MOD(@i,3) == 0 THEN SET @output = "Fizz"  
ELSE SET @output = @i  
ENDIF  
Output(Concat(@output,"<br>"))  
NEXT @i]%%
```



Answers

4

```
<table border="1">
%%[FOR @i = 0 TO 9 DO]%%
  <tr>
    %%[FOR @x = 1 TO 10 DO]%%
      <td>
        %%[SET @num = Add(Multiply(@i,10),@x)
        IF MOD(@num,15) == 0 THEN SET @output = "FizzBuzz"
        ELSEIF MOD(@num,5) == 0 THEN SET @output = "Buzz"
        ELSEIF MOD(@num,3) == 0 THEN SET @output = "Fizz"
        ELSE SET @output = @num
        ENDIF]%%
        %%=v(@output)=%%
      </td>
    %%[NEXT @x]%%
  </tr>
%%[NEXT @i]%%
</table>
```



Answers

5

```
<table border="1">
%%[FOR @i = 0 TO 9 DO]%%
  <tr>
    %%[FOR @x = 1 TO 10 DO
      SET @num = Add(Multiply(@i,10),@x)
      IF MOD(@num,15) == 0 THEN SET @bgcolor = "Blue"
      ELSEIF MOD(@num,5) == 0 THEN SET @bgcolor = "Green"
      ELSEIF MOD(@num,3) == 0 THEN SET @bgcolor = "Red"
      ELSE SET @bgcolor = "White"
      ENDIF]%%
      <td height="20px" width="20px"
        bgcolor="%%=v(@bgcolor)=%%">%%=v(@num)=%%</td>
    %%[NEXT @x]%%
  </tr>
%%[NEXT @i]%%
</table>
```

