

Salesforce Marketing Cloud

AMPscript Lookups & Loops 1 of 2

Exercise 3a



Overview

This exercise is designed to test your understanding of Data Extension Lookups and For Loops using Cloud Pages.

Process the tasks in order to complete this exercise.

Each Task in the following Brief will get progressively harder, so if you get stuck or are unsure how to proceed, you can join the conversation in the comments section here: <https://youtu.be/7Hi01EjGEKk>

The last pages of this document contain some solutions. Make sure you try solve them yourself before jumping to the end.

This is the first in a 2-part exercise. Try AMP-E03b after completing these tasks.

I hope these tasks give you some inspiration for creating dynamic Lookups and For Loops in your communications!

Required Files

This exercise requires the use of some Data Extensions with sample data.

You can copy/download all of the items you need from this exercise from this GitHub resource:

<https://github.com/camrobert/SalesforceMarketingCloud/tree/main/Challenge/AMP-E03>



Brief

You work in an online cloud education firm in the customer loyalty department. Your goal is to engage with learners on your platform by exciting them about their progress and other courses they could study.

Cloud Page Setup

- 1 In Web Studio/Cloud Pages, Create a new Cloud Page using the "Landing Page" type. Name your Cloud Page "AMP-E03 <yourname>".

To make this exercise easier to complete, all tasks will be conducted in this newly created, blank Cloud Page.

Note that you only need to click the "Schedule/Publish" to "preview" your results - you will not need to actually publish the page.

Some of the tasks in this exercise are not suitable to run on large data sets, and some tasks would normally be achieved using SQL and Reports. The goal of this exercise is to give you practical experience using a combination of HTML & AMPscript functions that can be translated into real world emails and Cloud Pages.

Hopefully this doesn't break the immersion too much and you still get the maximum value from the exercise.



Task 1

You work for an online learning platform, CloudLearners.
Your leader wants you to create a quick report to show the names of all the current/active Learners on the platform using a Cloud Page.

They would like the data to be in a Table structure (with a 1px black border) so that it is ordered and easy to read.

Objective

Create a new HTML content block on the Cloud Page, and use HTML & AMPscript to create a table with a column called "Name", outputting the name of all of the active learners.

Expected Result

Name
Alpha
Bravo
Charlie
Delta
Echo



Task 2

Before you can show your leader your progress, they have asked for you to include a Count of how many courses (Beams) the learner has completed, and a Sum of how many points they have earned on the platform.

Objective

Amend your table to include the "Beams" and "Points" columns.

Note: Check the Data Extension to understand how the data is structured. A course is only completed with the "CompleteDate" and "Points" fields of the [AMP-E03_CloudLearner_Beams] Data Extension contain values.

Expected Result

Name	Beams	Points
Alpha	10	33400
Bravo	10	33200
Charlie	6	13150
Delta	9	35100
Echo	3	14800



Task 3

Your leader loves the data so far!

The Chief Learning Officer has heard about your report and has asked for the name of the "Last Beam Completed" to be added to the report.

Objective

Amend your table to include the "Last Beam Completed" name.

Note: Check the Data Extension to understand how the data is structured.
You will need to Lookup the Name from the [AMP-E03_Beams] Data Extension.

Expected Result

Name	Beams	Points	Last Beam Completed
Alpha	10	33400	Designing for Services in the Cloud
Bravo	10	33200	Build your Cloud business
Charlie	6	13150	Cloud Admin Intermediate
Delta	9	35100	Designing for NonProfit in the Cloud
Echo	3	14800	Learn to think like the Cloud



Task 4

Everyone liked that!

The Chief Loyalty Officer is getting in on the action and would like you to add a column for "Unfinished Beams" to show a list of all the courses that each learner has started but not completed.

Objective

Amend your table to include a column called "Unfinished Beams", listing the names of each of the courses with no "CompleteDate" or "Points".

Expected Result

Name	Beams	Points	Last Beam Completed	Unfinished Beams
Alpha	10	33400	Designing for Services in the Cloud	
Bravo	10	33200	Build your Cloud business	Cloud Process Automation Beginner Cloud Admin Beginner
Charlie	6	13150	Cloud Admin Intermediate	
Delta	9	35100	Designing for NonProfit in the Cloud	Agile Development in the Cloud Commerce using Cloud
Echo	3	14800	Learn to think like the Cloud	Cloud Integration Methods Accelerate your Cloud Marketing



Task 5

We're in the End Game now.

The CEO them-self has heard of your work and has a curious question they want answered:

Given Learners earn points every time they complete a Beam, they would like to know what accuracy each learner has achieved on their completed courses so far, represented as a whole Percentage (XX%).

They scribble down the following equation to calculate the percentage:

$$\frac{\text{Learner's Points}}{\text{Sum of Completed Beam Points}}$$

Objective

Amend your table to include a column called "Accuracy", giving each learners percentage for completed Beams so far using the method provided.

Expected Result

Name	Beams	Points	Last Beam Completed	Unfinished Beams	Accuracy
Alpha	10	33400	Designing for Services in the Cloud		74%
Bravo	10	33200	Build your Cloud business	Cloud Process Automation Beginner Cloud Admin Beginner	73%
Charlie	6	13150	Cloud Admin Intermediate		73%
Delta	9	35100	Designing for NonProfit in the Cloud	Agile Development in the Cloud Commerce using Cloud	80%
Echo	3	14800	Learn to think like the Cloud	Cloud Integration Methods Accelerate your Cloud Marketing	76%



Answers



Answers

Note: Many of these Tasks can be solved multiple ways using HTML ^ AMPscript. The Answers below are not exclusively correct, however they will work if copied into your email for this exercise.

Answers are shown in HTML for formatting purposes.

1

```
<table border="1">
  <tr><th>Name</th></tr>
%%[SET @row = LookupRows("AMP-E03_CloudLearners","Active",1)
FOR @i = 1 TO ROWCOUNT(@row) DO]%%
  <tr><td>%%=Field(ROW(@row,@i),"Name")=%%</td></tr>
%%[NEXT @i]%%
</table>
```



Answers

2

```
<table border="1">
  <tr><th>Name</th><th>Beams</th><th>Points</th></tr>
%%[SET @row = LookupRows("AMP-E03_CloudLearners","Active",1)
FOR @i = 1 TO ROWCOUNT(@row) DO
  SET @beams = 0
  SET @points = 0
  SET @rowb = LookupRows("AMP-
E03_CloudLearner_Beams","CloudLearnerID",Field(Row(@row,@i),"ID"))
  FOR @b = 1 TO ROWCOUNT(@rowb) DO
    IF NOT EMPTY(Field(Row(@rowb,@b),"Points")) THEN
      SET @beams = add(1,@beams)
      SET @points = add(@points, Field(Row(@rowb,@b),"Points"))
    ENDIF
  NEXT @b
]%%
  <tr>
    <td>%%=Field(Row(@row,@i),"Name")=%%</td>
    <td>%%=v(@beams)=%%</td>
    <td>%%=v(@points)=%%</td>
  </tr>
%%[NEXT @i]%%
</table>
```



Answers

3

```
<table border="1">
  <tr>
    <th>Name</th><th>Beams</th><th>Points</th><th>Last Beam Completed</th>
  </tr>
  %%[
SET @Rowset = LookupRows('AMP-E03_CloudLearners','Active','1')
FOR @i = 1 TO RowCount(@Rowset) DO
SET @rowbeams = LookupOrderedRows('AMP-
E03_CloudLearner_Beams',100,'CompleteDate Desc',
'CloudLearnerID',Field(Row(@Rowset,@i),'ID'))

SET @Points = 0
SET @Beams = 0
SET @LastBeam = ""
FOR @b = 1 TO ROWCOUNT(@rowbeams) DO
IF Field(Row(@rowbeams,@b),'Points') > 0 THEN

IF @b == 1 THEN
SET @LastBeam = Lookup('AMP-E03_Beams', 'Name', 'ID',
Field(Row(@rowbeams,@b),'BeamsID'))
ENDIF

SET @Points = add(@Points,Field(Row(@rowbeams,@b),'Points'))
SET @Beams = add(@Beams,1)
ENDIF
NEXT @b

]%%
  <tr>
    <td>%%=Field(Row(@Rowset,@i),'Name')=%%</td>
    <td>%%=v(@Beams)=%%</td>
    <td>%%=v(@Points)=%%</td>
    <td>%%=v(@LastBeam)=%%</td>
  </tr>
  %%[NEXT @i]%%
</table>
```



Answers

4

```
<table border="1">
  <tr>
    <th>Name</th><th>Beams</th><th>Points</th><th>Last Beam Completed</th>
<th>Unfinished Beams</th>
  </tr>
%%[
SET @Rowset = LookupRows('AMP-E03_CloudLearners','Active','1')
FOR @i = 1 TO RowCount(@Rowset) DO
SET @rowbeams = LookupOrderedRows('AMP-E03_CloudLearner_Beams',100,'CompleteDate
Desc', 'CloudLearnerID',Field(Row(@Rowset,@i),'ID'))

SET @Points = 0
SET @Beams = 0
SET @LastBeam = ""
SET @Unfinished = ""
FOR @b = 1 TO ROWCOUNT(@rowbeams) DO
IF Field(Row(@rowbeams,@b),'Points') > 0 THEN
  IF @b == 1 THEN
    SET @LastBeam = Lookup('AMP-E03_Beams', 'Name', 'ID', Field(Row(@rowbeams,@b),'BeamsID'))
  ENDIF

  SET @Points = add(@Points,Field(Row(@rowbeams,@b),'Points'))
  SET @Beams = add(@Beams,1)
ELSE
  SET @Unfinished = Concat(@Unfinished, Lookup('AMP-E03_Beams', 'Name', 'ID',
Field(Row(@rowbeams,@b),'BeamsID')), "<br>")
ENDIF
NEXT @b

]%%
<tr>
  <td>%%=Field(Row(@Rowset,@i),'Name')=%%</td>
  <td>%%=v(@Beams)=%%</td>
  <td>%%=v(@Points)=%%</td>
  <td>%%=v(@LastBeam)=%%</td>
  <td>%%=v(@Unfinished)=%%</td>
</tr>
%%[NEXT @i]%%
</table>
```



Answers

5

```
<table border="1">
  <tr><th>Name</th><th>Beams</th><th>Points</th><th>Last Beam Completed</th>
<th>Unfinished Beams</th><th>Accuracy</th></tr>
%%[SET @row = LookupRows("AMP-E03_CloudLearners","Active",1)
FOR @i = 1 TO ROWCOUNT(@row) DO
  SET @beams = 0
  SET @points = 0
  SET @maxpoints = 0
  SET @latestName = ""
  SET @Unfinished = ""
  SET @rowb = LookupOrderedRows("AMP-E03_CloudLearner_Beams",0,"CompleteDate
DESC","CloudLearnerID",Field(Row(@row,@i),"ID"))
  FOR @b = 1 TO ROWCOUNT(@rowb) DO
    IF @b == 1 THEN
      SET @latestName = Lookup("AMP-E03_Beams","Name","ID",Field(Row(@rowb,@b),"BeamsID"))
    ENDIF

    IF NOT EMPTY(Field(Row(@rowb,@b),"Points")) THEN
      SET @beams = add(1,@beams)
      SET @points = add(@points, Field(Row(@rowb,@b),"Points"))
      SET @maxpoints = add(@maxpoints, Lookup("AMP-
E03_Beams","Points","ID",Field(Row(@rowb,@b),"BeamsID")))
    ELSE
      SET @Unfinished = Concat(@Unfinished,Lookup("AMP-
E03_Beams","Name","ID",Field(Row(@rowb,@b),"BeamsID")), "<br>")
    ENDIF
  NEXT @b
  SET @Percent = FormatNumber(Divide(@points,@maxpoints),"P0")
]%%
<tr>
  <td>%%=Field(ROW(@row,@i),"Name")=%%</td>
  <td>%%=v(@beams)=%%</td>
  <td>%%=v(@points)=%%</td>
  <td>%%=v(@latestName)=%%</td>
  <td>%%=v(@Unfinished)=%%</td>
  <td>%%=v(@Percent)=%%</td>
</tr>
%%[NEXT @i]%%
</table>
```

